# $Id: PKGBUILD 160468 2012-06-02 10:25:53Z bluewind $
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=glade-perl
_realname=Gtk2-GladeXML
pkgver=1.007
pkgrel=6
pkgdesc="Gtk2-GladeXML perl bindings for glade 2.x"
arch=(i686 x86_64)
license=('LGPL')
url="http://gtk2-perl.sourceforge.net/"
makedepends=('perl-extutils-pkgconfig' 'perl-extutils-depends')
depends=('libglade' 'gtk2-perl' 'perl')
options=('!emptydirs')
source=(http://downloads.sourceforge.net/sourceforge/gtk2-perl/${_realname}-${pkgver}.tar.gz)
md5sums=('e6ca234e2a9f0221263acd2a593c583b')

build() {
  cd "${srcdir}/${_realname}-${pkgver}"
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd "${srcdir}/${_realname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
